```
################################################################################
# Worksheet: mttex2
################################################################################


sage: e0 = EllipticCurve('446d1');
sage: p =5
sage: e0.rank()
2
sage: e0
Elliptic Curve defined by y^2 + x*y  = x^3 - x^2 - 4*x + 4 over Rational Field
sage: e0.is_ordinary(p)
True
sage: e0.Np(p)
10
The number of points in the reduction of $E$ at $p$ is divisible by
$p$, so $p$ is an <i>anomalous</i> prime for $E$.
sage: lp = e0.padic_lseries(p)
sage: lpst = lp.series(5)
sage: QpT = lpst.parent()
sage: T = QpT.0
sage: lps = lpst +O(T^7)
sage: lps
(5 + 5^2 + 3*5^3 + O(5^4))*T^2 + (2*5 + 3*5^2 + 3*5^3 + O(5^4))*T^3 + (4*5^2 + 4*5^3 + O(5^4))*T^4 + (4*5 + 4*5^2 + O(5^3))*T^5
sage: e0.padic_E2(p,prec=10)
3*5 + 4*5^2 + 5^3 + 5^4 + 5^5 + 2*5^6 + 4*5^7 + 3*5^9 + O(5^10)
sage: reg = e0.padic_regulator(p)
sage: R = pAdicField(p,10)
sage: kg = log(R(1+p))
sage: reg = R(reg)*p^2/kg^2
sage: reg*kg^2
2*5 + 2*5^2 + 5^4 + 4*5^5 + 2*5^7 + 4*5^8 + 2*5^9 + O(5^10)
the regulator normalized to the choice of $1+p$ as a topological
generator of $1+p\,\mathbb{Z}_p$.
sage: e0.tamagawa_numbers()
[2, 1]
sage: e0.torsion_order()
1
sage: lpstar = lps[2]
sage: lpstar
5 + 5^2 + 3*5^3 + O(5^4)
This is the leading term of the $p$-adic $L$-function. As it is not a
unit the prime $p$ is said to be an <i>irregular</i> prime.
sage: eps = (1-1/lp.alpha(20))^2
sage: lpstar/eps/reg/e0.tamagawa_product()*(e0.torsion_order())^2
1 + O(5^3)
This is the analytic $p$-adic order of Sha(E/Q). In particular it
shows that $Sha(E/\mathbb{Q})(5)$ is trivial. We can conclude that the
main conjecture holds for this curve.<br>
The characteristic series of the Selmer group is of the form $T^2\cdot
f(T)$ for an irreducible polynomial $f$ of the form
$T^4 + \cdots + 5\cdot u$ for a unit $u$. Actually, we can verify that
$f(T) = (T+1)^5 -1$. So $E(\mathbb{Q}_{1})$ has  rank $3$ and
$E(\mathbb{Q}_{\infty}$ has still rank $3$ and
$Sha(E/\mathbb{Q}_{\infty})(5)$ is still finite, probably still
trivial.
```